
Introductionto VHDL November4, 1998

IntegratedMicroelectronicsEngineering,Module1

Introductionto VHDL
Part1: VHDL for Design

Ed Casas

VHDL is a Very-complex1 Hardware DescriptionLanguageusedto
designandsimulatedigital logic circuits.

Intr oduction

Let’s startwith a simpleexample– a type of circuit calledan example1
that hasone output signal (c) which is the AND function of two input
signals(a andb). Thefile example1.vhd containsthefollowing VHDL
description:

-- An AND gate

library ieee ;
use ieee.std_logic_1164.all;

entity example1 is port (
a, b: in std_logic ;
c: out std_logic ) ;

end example1 ;

architecture rtl of example1 is
begin

c <= a and b ;
end rtl ;

First someobservationsonVHDL syntax:

� VHDL is case-insensitive. Therearemany capitalizationstyles. I
preferall lower-case.Whichever styleyouuse,beconsistent.

� Everythingfollowing two dashes“--” onaline is acommentandis
ignored.

� Statementscanbe split acrossany numberof lines. A semicolon
endseachstatement.Indentationstylesvary but an“end” shouldbe
indentedthesameasits corresponding“begin”

� Entity andsignalnamesbegin with a letterfollowedby letters,digits
or underscore(“ ”) characters.

1Actually, theV standsfor VHSIC andVHSIC standsfor VeryHigh SpeedIC.

� the library anduse statementsneedto be placedbeforeeach
entity/architecturepair (moreon this later).

A VHDL descriptionhastwo parts:anentity partandanarchitecture
part.

The entity part definesthe input andoutputsignalsfor the device or
“entity” beingdesignedwhile thearchitecturepartdescribesthebehaviour
of theentity. Althoughit’spossibleto defineseveralarchitecturesfor each
entity, we’ll only defineoneandcall it rtl.

Eacharchitectureis madeup of oneor morestatements,all of which
“execute2” at thesametime (concurrently) andcontinuously. Concurrent
executionallowsusto describethebehaviour of hardware(whichoperates
concurrentlyby nature).

Thesinglestatementin thisexampleis asignalassignmentthatassigns
the valueof an expressionto the outputsignalc. Expressionsinvolving
std_logic signalscanusethe logical operatorsand, or, nand, nor,
andxor, all of whichhaveequalprecedenceandnot whichhasthehigh-
estprecedence.Parenthesescanbeusedto forceevaluationin a different
order.

Fromthis VHDL descriptiona programcalleda logic synthesizercan
generatea circuit thatbehavesthesameway astheVHDL description.In
this caseit’s not too surprisingthat the synthesizeroutputsthe following
circuit:

Exercise: Write the VHDL description of a half-adder, a circuit that computes

the sum, sum, and carry, carry, of two one-bit numbers, a and b.

Vectorsand Buses

We can usesignalsof type std logic vector, one-dimensionalar-
raysof std logic, to modelbuses.Thedeclarationincludestheallowed
rangeof indices,for example:

a : std_logic_vector (3 downto 0)

The indices of bit vectors are usually declaredto have decreasing
(downto) ratherthanincreasing(to) valuessothatthefirst (leftmost)bit
is the mostsignificantone. std logic vector constantsareformed

2Theresultinghardwaredoesn’t actually“execute”but thispointof view is usefulwhen
usingVHDL for simulation.

by enclosinganorderedsequenceof std logic valuesin doublequotes
(e.g."0010").

Substrings(“slices”) of vectorscanbeextractedby specifyinga range
in theindex expression(e.g.a(3 downto 2)) andvectorscanbecon-
catenatedusingthe& operator(e.g. a & b). The logical operators(e.g.
and) canbeappliedto bit_vectorsandoperateonabit-by-bit basis.

declared as x : std logic vector (7 downto 0) ;, left by one bit and

sets the least-significant bit to zero.

SelectedAssignment

Theselectedassignmentstatementmodelstheoperationof amultiplexer –
thevalueassignedis selectedfrom severalpossiblevaluesby acontrolling
expression.Usingstd_logic_vectorsanda selectedassignmentwe
canconvert a truth tableinto aVHDL description.Thefollowing example
describesa3-to-8decoder:

-- 3-to-8 decoder

library ieee ;
use ieee.std_logic_1164.all;

entity decoder is
port (
a, b, c : in std_logic ;
y : out std_logic_vector (7 downto 0) ) ;

end decoder ;�
�

�

architecture rtl of decoder is
signal abc : std_logic_vector (2 downto 0) ;

begin
abc <= a & b & c ;

with abc select y <=
"00000001" when "000",
"00000010" when "001",
"00000100" when "010",
"00001000" when "011",
"00010000" when "100",
"00100000" when "101",
"01000000" when "110",
"10000000" when others ;

end rtl ;

Thekeywordothers indicatesthedefault valueto assignwhennone
of the othervaluesmatchesthe selectionexpression.Alwaysincludean
others clause.

Notethatlocal signalscanbedeclaredbetweenarchitecture and
begin.
TheVHDL descriptionsynthesizesto:

Conditional Assignment

A conditionalassignmentstatementbehaveslike if/elsestatementin acon-
ventionalprogramminglanguagebut, like the selectedassignmentstate-
ment,it is alsoaconcurrent statement.Unlike theselectedassignment,the
expressionsaretestedin orderandonly the first valuewhosecontrolling
expressionis trueis assigned.For example,thefollowing circuit is a4-to-3
priority encoder:

-- 4-to-3 encoder

library ieee ;
use ieee.std_logic_1164.all ;

entity encoder is port (
b : in std_logic_vector (3 downto 0) ;
n : out std_logic_vector (2 downto 0) ) ;

end encoder ; �
�

�

architecture rtl of encoder is
begin

n <=
"100" when b(3) = ’1’ else
"011" when b(2) = ’1’ else
"010" when b(1) = ’1’ else
"001" when b(0) = ’1’ else
"000" ;

end rtl ;

andtheresultingschematicis:
�
�

�
�20

Exercise: If we had used a selected assignment statement, how many lines

would have been required in the selected assignment?

SequentialCir cuits

So far we have only coveredcombinationallogic circuits– circuitswith-
outmemory. A process statementis usedto describesequentialcircuits.
Theprocess statementis aconcurrentstatementthatcontainssequential
statements.Sequentialstatementsexecuteoneaftertheotherasin conven-
tional programminglanguages.The implicationsof this will be covered
in detail later – for designpurposeswe only needto know how to usea
processto generateflip-flopsandregisters.

As anexample,wecandescribeaD flip-flop in VHDL asfollows:
�

-- D Flip-Flop

library ieee ;
use ieee.std_logic_1164.all;

entity dff is
port (
d, clk : in std_logic ;
q : out std_logic ) ;

end dff ;�
�

�

architecture rtl of dff is
begin
process(clk)
begin

if clk’event and clk = ’1’ then
q <= d ;

end if ;
end process ;
end rtl ;

The expressionclk’event (pronounced“clock tick event”) is true
when the valueof clk haschangedsincethe last time the processwas
executed.Thesignalq is only assignedavaluewhenclk changesandthe
new valueis 1. Otherwisetheoutputretainsits previousvalue.

Not surprisingly, theresultof synthesizingthisdescriptionis aflip-flop
thatis loadedon therisingedgeof theclock input:

Exercise: What would we get if we replaced d and q with signals of type

std logic vector?

StateMachines
�
�

�

Sequentiallogic circuitsareexamplesof statemachines.Recallthatastate
machinecomputesits outputandchangesstatebasedon its currentstate
andits inputs:

�

input output

next
state

current
 state

combinational
      logic

register

clock

Statemachinesin VHDL areimplementedby usingselectedor condi-
tional assignmentsto computeoutputandnext statefrom thecurrentstate
andtheinputs.A processis usedto implementtheregisterthatdefinesthe
currentstate.

Forexample,consideraVHDL descriptionfor aswitchdebouncerwith
thefollowing statetransitiondiagram:

Thiscanbedescribedasfollows:

-- Switch Debouncer

library ieee ;
use ieee.std_logic_1164.all ;

entity debounce is
port (

raw : in std_logic ;
clean : out std_logic ;
clk : in std_logic ) ;

end debounce ;�
�

�

architecture rtl of debounce is
signal currents, nexts :

std_logic_vector (1 downto 0) ;
begin

-- combinational logic for next state
nexts <=

"00" when raw = ’0’ else
"01" when currents = "00" else
"10" ;

-- combinational logic for output
clean <= ’1’ when currents = "10" else ’0’ ;�

�
�

-- sequential logic
process(clk)
begin

if clk’event and clk = ’1’ then
currents <= nexts ;

end if ;
end process ;

end rtl ;

Thesynthesizedcircuit is:
�

Exercise: Identify the components in the schematic that were created (“in-

stantiated ”) by different parts of the VHDL code.

Signedand UnsignedTypes

We often want to treatvectorsof logic valuesasbinary valuesin two’s-
complementor unsignedrepresentations.By declaringsignalsto beof type
signed orunsignedwecanusethestandardarithmetic(+, -, *, /)
and comparison(>, >, <=, >=,=, /=) operators. However, only
combinationallogic will be generatedandit maynot be possibleto syn-
thesizemultiplicationor division.

We canuseunsigned signalsto build a statemachinethat imple-
mentsacounter:

-- 3-bit Counter

library ieee ;
use ieee.std_logic_1164.all ;
use ieee.std_logic_arith.all ;

entity counter is
port (

count_out : out unsigned (2 downto 0) ;
clk : in std_logic ) ;

end counter ; �
�

�

architecture rtl of counter is
signal count, nextcount : unsigned (2 downto 0) ;

begin
nextcount <= count + 1 ;

process(clk)
begin

if clk’event and clk=’1’ then
count <= nextcount ;

end if ;
end process ;

count_out <= count ;
end rtl ;

We usedbothcount andcount_out becausethevalueof a signal
of typeout can’t be“read” insidethearchitecture.

�

�39 Exercise: Write the architecture for a 16-bit adder with two signed inputs, a

and b and a signed output c.

Components,Packagesand Libraries

Whendesigningcomplex logic circuitsit’sdesirableto decomposethede-
signinto simplerparts.Eachof thesepartscanbewrittenandtestedsepa-
rately, perhapsby differentpeople.If thepartsaresufficiently generalthen
it mightalsobepossibleto re-usethemin futureprojects.

Re-usein VHDL is doneby creating“components.” A component
declaration,likeanentitydeclarationspecifiesthecomponent’s inputsand
outputs.

Althoughcomponentscanbedeclaredwithin anarchitecture,compo-
nent declarationsare usually saved in “packages”. A packagetypically
containsa set of componentsfor a particularapplication. Packagesare
themselvesstoredin “libraries”:

Library
�

component

component

Package
�

component

component

Package
�

To save componentdeclarationswe put themwithin apackage dec-
laration.Whenthisfile is compiled(“analyzed”)theinformationaboutthe
componentsin thepackageis savedin afile. Thecomponentsin thepack-
agecanthenbeusedin otherdesignsby makingthemvisible with ause
statement.

A componentdeclarationis similar to anentitydeclarationanddefines
the input andoutputsignals.Note thata componentdeclarationdoesnot
createhardware– only whenthecomponentis usedin anarchitecturewill
hardwarebegenerated(“instantiated”).

For example,thefollowing codecreatesapackagecalledflipflops
containingonly onecomponentcalledrs with inputsr ands andanout-
putq whenit is compiled

�
�

�

- packagenameis flipflops - declaresthers component:

package flipflops is
component rs

port ( r, s : in bit ; q : out bit ) ;
end component ;

end flipflops ;

- compilingthis file createsthepackage
To make the objectsin a packageavailable(“visible”) in anotherde-

sign,we uselibrary statementsto specifythe librariesto besearched
andause statementfor eachpackagewewishto use.Thetwo mostcom-
monlyusedlibrariesarecalledIEEE andWORK.

In mostVHDL implementationsa library is adirectoryandeachpack-
ageis a file in that directory. The packagefile is a databasecontaining
variousthingsincludinginformationaboutthecomponentsin thepackage
(thecomponentinputs,outputs,types,etc).

TheWORK library is alwaysavailablewithout having to usea library
statementandis the library into which designentitiesareplacedasthey
arecompiled.

library anduse statementsmustbe usedbeforeeach designunit
(entityor architecture)thatmakesuseof componentsfoundin thosepack-
ages.For example,if you wantedto usetheDSP packagein theALTERA
library youwoulduse:

�

library altera ;
use altera.dsp.all ;

UsingComponents

Oncea componenthasbeenplacedin a package,it canbeused(“instanti-
ated”)in anarchitecture.A componentinstantiationsimplydescribeshow
the componentis “hooked up” to the othersignalsin the architecture.It
is a concurrent statementlike the process statementratherthana se-
quential statementandsoacomponentinstantiationcannotbeput insidea
process.

�

The following exampleshows how 2-input exclusive-or gatescanbe
usedto built a 4-inputparity-checkcircuit usingcomponentinstantiation.
Thistypeof descriptionis calledstructural VHDL becausewearedefining
thestructureratherthanthebehaviour of thecircuit.

Thereare two files: the first file describesthexor2 component(al-
thougha typicalpackagedefinesmorethanonecomponent):

�
�

�

-- define an xor2 component in a package

library ieee ;
use ieee.std_logic_1164.all ;

package xor_pkg is
component xor2

port ( a, b : in std_logic ; x : out std_logic ) ;
end component ;

end xor_pkg ;

the secondfile describestheparity entity that usesthexor2 compo-
nent:

�

-- Parity function built from xor gates

library ieee ;
use ieee.std_logic_1164.all ;

use work.xor_pkg.all ;

entity parity is
port ( a, b, c, d : in std_logic ; p : out std_logic ) ;

end parity ;�
�

�
�46

architecture rtl of parity is
-- internal signals
signal x, y : std_logic ;

begin
x1: xor2 port map ( a, b, x ) ;
x2: xor2 port map ( c, x, y ) ;
x3: xor2 port map ( d, y, p ) ;

end rtl ;

Theresultingtop-level schematicfor theparity entity is:
�

Exercise: Label the connections within the parity generator schematic with

the signal names used in the architecture.

Notethatacomponentneednothave beencreatedin VHDL – it could
beanASIC standard-cellor FPGAconfigurationinformation.

TypeDeclarations

It’s oftenusefulto make up new typesof signalsfor a project.We cando
this in VHDL by usingtypedeclarations.Thetwo mostcommonusesfor

definingnew typesareto declarearraysof given dimensions(e.g. a bus
of a given width) andto declaretypesthat canonly have oneof a setof
possiblevalues(calledenumerationtypes).

�

Thefollowing exampleshowshow apackagecalleddsp_types that
declarestwo new typesis created:

package dsp_types is
type mode is (slow, medium, fast) ;
subtype sample is std_logic_vector (7 downto 0) ;

end dsp_types ;

Note that we needto usea subtype declarationin the secondex-
amplebecausethestd_logic_vector type is alreadydefined.Type
declarationsareoftenplacedin packagesto make themavailableto multi-
pledesignunits.

Tri-State Buses

A tri-stateoutput can be set to the normal high and low logic levels as
well asto a high-impedancestate.This typeof outputis oftenusedwhere
differentdevicesmustdriveabusatdifferenttimes.Oneway to specifyto
theVHDL synthesizerthatanoutputshouldbesetto thehigh-impedance
stateis to useasignalof std_logic typeandassignit avalueof ’Z’.

�
�

�
�51

Thefollowing exampleshowsanimplementationof a4-bit buffer with
an enableoutput. Whenthe enableis not assertedthe output is in high-
impedancemode:

�

-- Tri-State Buffer

library ieee ;
use ieee.std_logic_1164.all ;

entity tbuf is port (
d : in std_logic_vector (3 downto 0) ;
q : out std_logic_vector (3 downto 0) ;
en : in std_logic
) ;

end tbuf ; �
�

architecture rtl of tbuf is
begin

q <=
d when en = ’1’ else
"ZZZZ" ;

end rtl ;

Theresultingschematicfor thetbuf is:
�
�

�

FPGA Configuration Demonstration

To demonstratetheuseof VHDL we will designa “light chaser”by com-
biningthecounteranddecoderdescribedearlier. Thedecoder andcounter
entitiesareplacedin ademocom package(not shown) andinstantiatedin
thearchitectureof thedemo entity.

An internalsignalanda type-conversionfunctionareusedto convert
counter’s unsigned outputto thethreestd_logic inputsrequiredby
the decoder. A secondinternalsignal is usedto convert the active-high
decoderoutputto therequiredactive-low LED drive signals.

The FPGA demonstrationboardis configuredso that theclk signal
comesfrom a pushbutton and the led outputsdrive the individual seg-
mentsof a7-segmentLED display.

-- demonstration design light chaser

library ieee ;
use ieee.std_logic_1164.all;
use ieee.std_logic_arith.all;
use work.democom.all ;

entity demo is
port (
clk : in std_logic ;
led : out std_logic_vector (7 downto 0) ) ;

end demo;�
�

�

architecture rtl of demo is
signal count : unsigned (2 downto 0) ;
signal scount : std_logic_vector (2 downto 0) ;
signal ledN : std_logic_vector(7 downto 0) ;

begin
c1: counter port map ( count, clk ) ;
scount <= std_logic_vector(count);

d1: decoder port map ( scount(2), scount(1), scount(0), ledN ) ;
led <= not ledN ;

end rtl ;

